

Errata for QuadCrawler Manual v3.1 (#30073)

Distance Sensor Upgrade

Your QuadCrawler robot kit's distance sensor has been upgraded to a PING))) Ultrasonic Distance Sensor(part #28015). The complete product documentation and example programs can be downloaded from the PING))) Sensor's product page: http://www.parallax.com/detail.asp?product_id=28015


Errata

Page 1: Parts List: replace the GP2D12 with:

- (1) PING))) Ultrasonic Distance Sensor
- (1) Servo Extension Cable
- (1) 3-pin male/male header

Page 63: replace with the following:

Your QuadCrawler kit is equipped with the Parallax PING))) Ultrasonic Distance Sensor. Use the Servo Extension Cable, 3-pin header, and jumper wires provided in your kit to connect the PING))) sensor to the Board of Education breadboard, following the schematic in Figure 54.



Pages 96-97: replace with the following:

This code, adapted from the PING))) sensor product documentation, is used to obtain distance readings.

```

' =====
' File..... PingDemoQuad.BS2
' Purpose.... Demo Code for Parallax Ping Sonar Sensor
' Author..... Parallax, Inc.
' Updated.... 28 MAR 2006
'
'   {$STAMP BS2}
'   {$PBASIC 2.5}
' =====
' -----[ Program Description ]-----
'
' This program demonstrates the use of the Parallax Ping Sonar sensor and
' converting the raw measurement to English (inches) and Metric (cm) units.
'
' Sonar Math:
' At sea level sound travels through air at 1130 feet per second. This
' equates to 1 inch in 73.746 uS, or 1 cm in 29.034 uS).
'
' Since the Ping sensor measures the time required for the sound wave to
' travel from the sensor and back. The result -- after conversion to
' microseconds for the BASIC Stamp module in use -- is divided by two to
' remove the return portion of the echo pulse. The final raw result is
' the duration from the front of the sensor to the target in microseconds.
'
' -----[ I/O Definitions ]-----
Ping                PIN                14
' -----[ Constants ]-----
    
```

```

Trigger          CON      5          ' trigger pulse = 10 uS
Scale            CON     $200        ' raw x 2.00 = uS

RawToIn         CON     889          ' 1 / 73.746 (with **)
RawToCm         CON    2257          ' 1 / 29.034 (with **)

IsHigh          CON      1          ' for PULSOUT
IsLow           CON      0

' -----[ Variables ]-----
rawDist         VAR     Word          ' raw measurement
inches          VAR     Word
cm              VAR     Word

' -----[ Initialization ]-----

Reset:
  DEBUG CLS,          ' setup report screen
    "Parallax Ping Sonar ", CR,
    "=====", CR,
    CR,
    "Time (uS)..... ", CR,
    "Inches..... ", CR,
    "Centimeters... "

' -----[ Program Code ]-----

Main:
  DO
    GOSUB Get_Sonar          ' get sensor value
    inches = rawDist ** RawToIn ' convert to inches
    cm = rawDist ** RawToCm   ' convert to centimeters

    DEBUG CRSRXY, 15, 3,          ' update report screen
      DEC rawDist, CLREOL,
      CRSRXY, 15, 4,
      DEC inches, CLREOL,
      CRSRXY, 15, 5,
      DEC cm, CLREOL

    PAUSE 100
  LOOP
END

' -----[ Subroutines ]-----
' This subroutine triggers the Ping sonar sensor and measures the echo pulse. The
' raw value from the sensor is converted to microseconds. This value is divided
' by two to remove the return trip - the result value is the distance from the
' sensor to the target in microseconds.

Get_Sonar:
  Ping = IsLow          ' make trigger 0-1-0
  PULSOUT Ping, Trigger ' activate sensor
  PULSIN Ping, IsHigh, rawDist ' measure echo pulse
  rawDist = rawDist * / Scale ' convert to uS
  rawDist = rawDist / 2 ' remove return trip
RETURN

```